
Recomendación UIT-T G.107 

El modelo E, un modelo informático para utilización 
en planificación de la transmisión 

 

 

 

Resumen 
La presente Recomendación describe el algoritmo para el denominado modelo E como el modelo 
común del UIT-T para la determinación de índices de transmisión. Este modelo informático puede 
resultar útil a los planificadores de sistemas de transmisión para contribuir a que los usuarios estén 
satisfechos de la calidad de la transmisión de extremo a extremo. El resultado primario del modelo es 
una cuantificación escalar de la calidad de transmisión. Una característica fundamental de este 
modelo es la utilización de factores de degradación de la transmisión que reflejen los efectos de los 
modernos dispositivos de procesamiento de señales. 

En la revisión efectuada en el año 2000, se proporcionó una versión mejorada del modelo E con el 
fin de tener en cuenta mejor los efectos del ruido ambiente en el lado emisor y la distorsión de 
cuantificación. En la revisión de 2002 se ha incluido la degradación debida a la pérdida aleatoria de 
paquetes en una manera paramétrica para diferentes códecs. En relación con la versión de 2003, 
puede verse un modelado mejorado de la calidad para niveles bajos del efecto local del hablante. La 
presente revisión permite realizar predicciones más precisas sobre la calidad de los códecs en 
condiciones de pérdida de paquetes dependiente (a corto plazo). 

 

 

Orígenes 

La Recomendación UIT-T G.107 fue aprobada el 1 de marzo de 2005 por la Comisión de Estudio 12 
(2005-2008) del UIT-T por el procedimiento de la Recomendación UIT-T A.8. 

PREFACIO 

La UIT (Unión Internacional de Telecomunicaciones) es el organismo especializado de las Naciones Unidas 
en el campo de las telecomunicaciones. El UIT-T (Sector de Normalización de las Telecomunicaciones de la 
UIT) es un órgano permanente de la UIT. Este órgano estudia los aspectos técnicos, de explotación y 
tarifarios y publica Recomendaciones sobre los mismos, con miras a la normalización de las telecomunica-
ciones en el plano mundial. 

La Asamblea Mundial de Normalización de las Telecomunicaciones (AMNT), que se celebra cada cuatro 
años, establece los temas que han de estudiar las Comisiones de Estudio del UIT-T, que a su vez producen 
Recomendaciones sobre dichos temas. 

La aprobación de Recomendaciones por los Miembros del UIT-T es el objeto del procedimiento establecido 
en la Resolución 1 de la AMNT. 

En ciertos sectores de la tecnología de la información que corresponden a la esfera de competencia del 
UIT-T, se preparan las normas necesarias en colaboración con la ISO y la CEI. 

 

 

 

NOTA 

En esta Recomendación, la expresión "Administración" se utiliza para designar, en forma abreviada, tanto 
una administración de telecomunicaciones como una empresa de explotación reconocida de 
telecomunicaciones. 

La observancia de esta Recomendación es voluntaria. Ahora bien, la Recomendación puede contener ciertas 
disposiciones obligatorias (para asegurar, por ejemplo, la aplicabilidad o la interoperabilidad), por lo que la 
observancia se consigue con el cumplimiento exacto y puntual de todas las disposiciones obligatorias. La 
obligatoriedad de un elemento preceptivo o requisito se expresa mediante las frases "tener que, haber de, hay 
que + infinitivo" o el verbo principal en tiempo futuro simple de mandato, en modo afirmativo o negativo. El 
hecho de que se utilice esta formulación no entraña que la observancia se imponga a ninguna de las partes. 

 

 

 

PROPIEDAD INTELECTUAL 

La UIT señala a la atención la posibilidad de que la utilización o aplicación de la presente Recomendación 
suponga el empleo de un derecho de propiedad intelectual reivindicado. La UIT no adopta ninguna posición 
en cuanto a la demostración, validez o aplicabilidad de los derechos de propiedad intelectual reivindicados, 
ya sea por los miembros de la UIT o por terceros ajenos al proceso de elaboración de Recomendaciones. 

En la fecha de aprobación de la presente Recomendación, la UIT no ha recibido notificación de propiedad 
intelectual, protegida por patente, que puede ser necesaria para aplicar esta Recomendación. Sin embargo, 
debe señalarse a los usuarios que puede que esta información no se encuentre totalmente actualizada al 
respecto, por lo que se les insta encarecidamente a consultar la base de datos sobre patentes de la TSB. 

 

 

 

  UIT  2005 

Reservados todos los derechos. Ninguna parte de esta publicación puede reproducirse por ningún 
procedimiento sin previa autorización escrita por parte de la UIT. 

Recomendación UIT-T G.107 

El modelo E, un modelo informático para utilización 
en planificación de la transmisión 

1 Generalidades 

1.1 Alcance 
La presente Recomendación describe un modelo informático, conocido como modelo E, que ha 
demostrado ser útil como herramienta de planificación de sistemas de transmisión, para evaluar los 
efectos combinados de las variaciones de diversos parámetros de transmisión que afectan a la 
calidad de la conversación1 telefónica con microteléfono a 3,1 kHz. Este modelo informático puede 
ser utilizado, por ejemplo, por los planificadores de sistemas de transmisión para asegurar que los 
usuarios están satisfechos con las características de transmisión de extremo a extremo, a la vez que 
se evita el sobredimensionamiento de las redes. Hay que destacar que el resultado fundamental del 
modelo es el "factor de determinación de índices" R, aunque se puede transformar para obtener 
estimaciones de la opinión de los clientes. Estas estimaciones se hacen solamente a los efectos de la 
planificación de la transmisión y no para predecir la opinión real de los clientes (para lo cual no hay 
un modelo acordado recomendado por el UIT-T). 

La presente revisión incluye la pérdida de paquetes como un nuevo parámetro y la mejora del 
modelo del efecto local del hablante. 

El modelo E no ha sido verificado totalmente mediante mediciones en servicio real o pruebas de 
laboratorio para  el gran número de posibles combinaciones de parámetros de entrada. En el caso de 
muchas combinaciones que son muy importantes para los planificadores de sistemas de transmisión, 
el modelo E se puede utilizar con confianza; pero para otras combinaciones de parámetros, las 
predicciones del modelo E han sido cuestionadas y se encuentran actualmente en estudio. En 
consecuencia, se ha de tener cuidado al utilizar el modelo E para determinadas condiciones; por 
ejemplo, el modelo E puede dar resultados inexactos para las combinaciones de ciertos tipos de 
degradaciones. El anexo A proporciona más información a este respecto. 

1.2 Referencias 

Las siguientes Recomendaciones del UIT-T y otras referencias contienen disposiciones que, 
mediante su referencia en este texto, constituyen disposiciones de la presente Recomendación. Al 
efectuar esta publicación, estaban en vigor las ediciones indicadas. Todas las Recomendaciones y 
otras referencias son objeto de revisiones por lo que se preconiza que los usuarios de esta 
Recomendación investiguen la posibilidad de aplicar las ediciones más recientes de las 
Recomendaciones y otras referencias citadas a continuación. Se publica periódicamente una lista de 
las Recomendaciones UIT-T actualmente vigentes. En esta Recomendación, la referencia a un 
documento, en tanto que autónomo, no le otorga el rango de una Recomendación. 

[1] Recomendación UIT-T G.100 (2001), Definiciones utilizadas en las Recomendaciones 
sobre características generales de las conexiones y circuitos telefónicos internacionales. 

[2] Recomendación UIT-T G.108 (1999), Aplicación del modelo E: Directrices para la 
planificación. 

____________________ 
1  En este contexto, la calidad de la conversación hace referencia a las características de transmisión, por 

ejemplo, tiempos de transmisión prolongados, efectos del eco para el hablante, etc. No obstante, el 
modelo E descrito en esta Recomendación no está destinado a modelar las degradaciones de transmisión 
en situaciones de conversación simultánea. 

[3] Recomendación UIT-T G.109 (1999), Definición de las categorías de calidad de 
transmisión vocal. 

[4] Recomendación UIT-T G.113 (2001), Degradaciones de la transmisión debido al 
tratamiento de las señales vocales. 

[5] Recomendación UIT-T G.113 apéndice I (2002), Valores provisionales de planificación 
para el factor de degradación de equipo, Ie, y el factor de robustez contra pérdida de 
paquetes, Bpl. 

[6] Recomendación UIT-T P.833 ((2001), Metodología para la obtención de los factores de 
degradación del equipo a partir de pruebas subjetivas de escucha solamente. 

[7] Recomendación UIT-T P.834 (2002), Metodología para la obtención de los factores de 
degradación del equipo a partir de modelos instrumentales. 

[8] Recomendación UIT-T P.862 (2001), Evaluación de la calidad vocal por percepción: Un 
método objetivo para la evaluación de la calidad vocal de extremo a extremo de redes 
telefónicas de banda estrecha y códecs vocales. 

2 Modelo E, un modelo informático para uso en planificación de la transmisión 

2.1 Introducción 
La complejidad de las redes modernas requiere que, para la planificación de transmisión, los 
numerosos parámetros de transmisión no sean considerados sólo individualmente, sino que también 
se tengan en cuenta sus efectos combinados. Aunque esto se puede hacer mediante "estimación 
basada en la experiencia" es deseable un planteamiento más sistemático, como la utilización de un 
modelo informático. El resultado del modelo descrito aquí es un valor escalar de determinación de 
índice de calidad, R, que varía linealmente con la calidad global de la conversación. La 
Rec. UIT-T G.113 [4] contiene orientaciones sobre las degradaciones específicas, incluidos los 
efectos combinados, basándose en una simplificación del modelo. Sin embargo, estos resultados 
pueden dar también estimaciones nominales de la reacción de los usuarios, por ejemplo en forma de 
porcentajes que consideran la conexión modelada "buena o mejor" o "mediocre o peor", como se 
describe en el anexo B. Además, en la Rec. UIT-T G.108 [2] se proporciona orientación detallada 
para la aplicación correcta del modelo E, descrito en la presente Recomendación, mientras que en la 
Rec. UIT-T G.109 [3] se definen las categorías de calidad de transmisión vocal. 

2.2 Código fuente 
El anexo C incluye el código fuente en BASIC del modelo E que se describe en la presente 
Recomendación. El objeto de este código es asegurar que los usuarios del modelo E están utilizando 
implementaciones coherentes de las fórmulas. 

3 Estructura y algoritmos básicos del modelo E 
El modelo E se basa en el método de factor de degradación del equipo, de acuerdo con modelos de 
determinación de índices de transmisión previos. Fue elaborado por un Grupo ad hoc del ETSI 
denominado "Calidad de transmisión de la voz  de boca a oído". 

Como se muestra en la figura 1, la conexión de referencia se divide entre un lado transmisión y un 
lado recepción. El modelo estima la calidad de comunicación de la conversación de boca a oído 
percibida por el usuario en el lado recepción, como oyente y como hablante. 

G.107_F01

Distorsión de cuantificación qdu 

Factor de mejora A 

Retardo medio en un sentido T  

Retardo absoluto Ta

SLR RLR

OLR

Punto de 0 dBr

Factor  Ds

Ruido de circuito Nc 
referido a 0 dBr

Ruido
ambiente Ps 

Pérdida de trayecto de eco
ponderada WEPL 

Retardo de ida
y vuelta Tr 

Lado emisión Lado recepción

Índice de efecto
local del oyente
LSTR (LSTR =
STMR + Dr)

Índice de sonoridad
del eco para el
hablante TELR

Índice de 
enmascaramiento
para el efecto
local STMR

Ruido
ambiente Pr

Factor Dr

Factor de degradación del equipo Ie
Factor de robustez contra pérdida
de paquetes Bpl

Codificación/decodificación

Probabilidad de pérdida de paquetes Ppl

 

Figura 1/G.107 − Conexión de referencia del modelo E 

Los parámetros de transmisión utilizados como entrada para el modelo informático se muestran en 
la figura 1. En el algoritmo se tratan por separado los valores de ruido ambiente y de los factores D 
para el lado emisión y el lado recepción y pueden tener valores diferentes. Los parámetros SLR, 
RLR y ruido de circuito Nc están referidos a un punto definido de 0 dBr. Los demás parámetros de 
entrada se consideran como valores para la conexión global, por ejemplo, OLR (en cualquier caso la 
suma de SLR y RLR), número de qdu, factores de degradación del equipo Ie y factor de mejora A, o 
se refieren únicamente al lado recepción, como STMR, LSTR, WEPL (para el cálculo del eco para 
el oyente) y TELR. 

Hay tres parámetros diferentes asociados con el tiempo de transmisión. El retardo absoluto Ta 
representa el retardo total en un sentido entre el lado emisión y el lado recepción y se utiliza para 
estimar la degradación debida a retardos demasiado largos. El parámetro retardo medio en un 
sentido T representa el retardo entre el lado recepción (en estado hablante) y el punto de una 
conexión en el que aparece un acoplamiento de señales como una fuente de eco. El retardo de ida y 
vuelta Tr sólo representa el retardo en un bucle a 4 hilos en el que la señal "doblemente reflejada" 
producirá degradaciones debidas al eco para el oyente. 

3.1 Cálculo del factor de determinación de índices de transmisión, R 
Según el método del factor de degradación del equipo, el principio fundamental del modelo E se 
basa en un concepto dado en la descripción del modelo OPINE [véase la Bibliografía, Suplemento 3 
a las Recomendaciones de la serie P]: 

Los factores sicológicos y la escala sicológica son aditivos. 

En una primera etapa, el resultado de cualquier cálculo con el modelo E es un factor de 
determinación de índices de transmisión R, que combina todos los parámetros de transmisión 
pertinentes para la conexión considerada. Este factor R está constituido por: 

  AIe-effIdIsRoR +−−−=  (3-1) 

Ro representa en principio la relación señal/ruido básica que incluye fuentes de ruido, tales como 
ruido de circuito y ruido ambiente. El factor Is es una combinación de todas las degradaciones que 
aparecen de forma más o menos simultánea con la señal vocal. El factor Id representa las 
degradaciones producidas por el retardo y el factor de degradación efectiva del equipo Ie-eff 
representa las degradaciones producidas por códecs de velocidad binaria baja. Incluye también la 
degradación debida a pérdidas de paquetes de distribución aleatoria. El factor de mejora A permite 
compensar los factores de degradación cuando existen otras ventajas de acceso para el usuario. El 
término Ro y los valores Is e Id se subdividen en valores de degradación específicos más detallados. 
Las cláusulas siguientes muestran las fórmulas utilizadas en el modelo E. 

3.2 Relación señal/ruido básica, Ro 
La relación señal/ruido básica Ro está definida por: 

  ( )NoSLRRo +−= 5,115  (3-2) 

El término No [en dBm0p] es la suma de las potencias de diferentes fuentes de ruido: 

  











+++= 10101010 10101010log10

NfoNorNosNc

No  (3-3) 

Nc [en dBm0p] es la suma de todas las potencias de ruido de circuito, referidas al punto de 0 dBr. 

Nos [en dBm0p] es el ruido de circuito equivalente en el punto de 0 dBr, producido por el ruido 
ambiente Ps en el lado emisión: 

  ( )214004,0100 −−−+−−−= DsOLRPsDsSLRPsNos  (3-4) 

donde OLR = SLR + RLR. De igual manera, el ruido ambiente Pr en el lado recepción se transforma 
en un ruido de circuito equivalente Nor [en dBm0p] en el punto de 0 dBr. 

  2)35(008,0121 −++−= PrePreRLRNor  (3-5) 

El término Pre [en dBm0p] es el "ruido ambiente efectivo" producido por la amplificación de Pr en 
el trayecto de efecto local del oyente: 

  











++= 10

)–10(

101log10
LSTR

PrPre  (3-6) 

Nfo [en dBm0p] representa el "nivel de ruido" en el lado recepción, 

  RLRNforNfo +=  (3-7) 

con Nfor fijado normalmente en –64 dBmp. 

3.3 Factor de degradaciones simultáneas, Is 
El factor Is es la suma de todas las degradaciones que pueden producirse de forma más o menos 
simultánea con la transmisión de señales vocales. El factor Is se divide en tres factores de 
degradación específicos: 

  IqIstIolrIs ++=  (3-8) 

Iolr representa la disminución de calidad producida por valores demasiado bajos de OLR y viene 
dada por: 

  





































+=

8
–

8
120

8
1

8 XolrXolrIolr  (3-9) 

donde: 

  )64(2,0 RLRNoOLRXolr −++=  (3-10) 

El factor Ist representa la degradación producida por efectos locales no óptimos: 

29
33

3–113
4,19

1128–
6

13–112
13
1

1335
1

358
1

8
+


















+−



















 ++

















+= STMRoSTMRoSTMRoIst  (3-11) 

donde: 

  











+=

−−−
10410 10e10log10–

TELRTSTMR

STMRo  (3-12) 

El factor de degradación Iq representa la degradación producida por la distorsión de cuantificación: 

  [ ]ZYIq 10101log15 ++=  (3-13) 

donde: 

  
94,8

46
15

100 GRY o −+−=  (3-14) 

  
4030

46 GZ −=  (3-15) 

y: 

  20602,0258,007,1 QQG ++=  (3-16) 

  )qdu(log1537 −=Q  (3-17) 

En esta fórmula, qdu representa el número de qdu para toda la conexión entre el lado emisión y el 
lado recepción. 
NOTA – Si se utiliza un factor de degradación Ie para una pieza del equipo, no se debe utilizar el valor qdu 
para esa misma pieza del mismo equipo. 

3.4 Factor de degradación por retardo, Id 
El factor de degradación Id, que representa todas las degradaciones debidas al retardo de las señales 
vocales, también se subdivide en tres factores Idte, Idle e Idd: 

  IddIdleIdteId ++=  (3-18) 

El factor Idte expresa una estimación para las degradaciones debidas al eco para el hablante: 

  ( )TReRoeReRoeIdte –
2

e–11–100
4

)–(
2
–












++=  (3-19) 

donde: 

  )(5,1 RLRNoRoe −−=  (3-20) 

  )14(5,280 −+= TERVRe  (3-21) 

  
23,0–e6

150
1

10
1

log40– T
T

T

TELRTERV +
+

+
=  (3-22) 

Para valores de T < 1 ms, el eco para el hablante debe ser considerado como efecto local, es decir, 
Idte = 0. El algoritmo de cálculo combina además la influencia de STMR en el eco para el hablante. 
Teniendo en cuenta que los valores bajos de STMR pueden tener cierto efecto de enmascaramiento 
en el eco para el hablante y que para valores muy altos de STMR el eco para el hablante puede ser 
más perceptible, los términos TERV e Idte se ajustan como sigue: 

Para STMR < 9 dB: 

Se sustituye TERV en la ecuación 3-21 por TERVs, donde: 

  
2

IstTERVTERVs +=  (3-23) 

Para 9 dB ≤ STMR ≤ 20 dB: 

se aplican las ecuaciones 3-19 a 3-22 indicadas anteriormente. 

Para STMR > 20 dB: 

En la ecuación 3-18, Idte se sustituye por Idtes, donde: 

  22 IstIdteIdtes +=  (3-24) 

El factor Idle representa degradaciones debidas al eco para el oyente. Las ecuaciones son: 

  169
4

)–(
2
– 2

++= RleRoRleRoIdle  (3-25) 

donde: 

  25,0–)1)(7(5,10 ++= TrWEPLRle  (3-26) 

El factor Idd representa la degradación producida por retardos absolutos demasiado largos Ta, que 
se producen incluso con compensación perfecta del eco. 

Para Ta ≤ 100 ms: 
  0=Idd  

Para Ta > 100 ms: 

  ( )


















+

















++= 2

3
13–125

6
1

6
6
1

6 XXIdd  (3-27) 

con: 

  
2log

100
log 








=

Ta

X  (3-28) 

3.5 Factor de degradación del equipo, Ie 
Los valores para el factor de degradación del equipo, Ie, de elementos que utilizan códecs de baja 
velocidad binaria no están relacionados con otros parámetros de entrada. Dependen de resultados de 
pruebas subjetivas con notas medias de opinión, así como de la experiencia de la red. Para los 
valores de Ie efectivamente recomendados, véase el apéndice I/G.113 [5]. 

Los valores específicos del factor de degradación para el funcionamiento de códecs en condiciones 
de pérdida de paquetes aleatoria2 han sido tratados anteriormente utilizando los valores Ie 
tabulados, dependientes de la pérdida de paquetes. Ahora se define el factor de robustez contra 
pérdida de paquetes,  Bpl,  como un valor específico del códec. El factor de degradación efectiva del 
equipo que depende de la pérdida de paquetes, Ie-eff, se obtiene utilizando el valor específico del 
códec para el factor de degradación del equipo con pérdida de paquetes cero, Ie, y el factor de 
robustez contra pérdida de paquetes, Bpl, ambos enumerados en el  apéndice I/G.113 para varios 
códecs. Con la probabilidad de pérdida de paquetes Ppl, Ie-eff se calcula usando la fórmula: 

  
Bpl

BurstR
Ppl

PplIeIeIe-eff
+

⋅−+= )95(  (3-29) 

BurstR es la así llamada relación de ráfaga, que se define como sigue:  

"arbitraria" pérdida de scondicioneen  red laen  previstas ráfagas las de media Longitud
llegada de secuencia unaen  observadas ráfagas las de media Longitud=BurstR  

Cuando la pérdida de paquetes sea aleatoria (esto es, independiente), BurstR = 1; y 
cuando la pérdida de paquetes obedezca a las ráfagas (es decir, sea dependiente), BurstR > 1. 

Por ejemplo, tratándose de distribuciones de la pérdida de paquetes correspondientes a un modelo 
de Markov con dos estados y las siguientes probabilidades de transición: p entre un estado de 
paquetes "localizados" y otro de "pérdida", y q entre un estado de "pérdida" y otro de paquetes 
"localizados", la relación de ráfaga es la siguiente: 

  
q

Ppl
p

Ppl
qp

BurstR 100/1100/1 −==
+

=  (3-30) 

De acuerdo con la fórmula 3-29, el factor de degradación efectiva del equipo en caso de Ppl = 0 (no 
hay pérdida de paquetes) es igual al valor Ie definido en el apéndice I/G.113. 

Sírvanse consultar el anexo A/G.107 en lo que respecta a la gama de valores de los parámetros para 
la cual se ha validado el algoritmo. 

3.6 Factor de mejora, A 
Debido al significado específico del factor de mejora A, no existe, por consiguiente, ninguna 
relación con los demás parámetros de transmisión. El cuadro 1 ofrece algunos valores provisionales. 

____________________ 
2  Se considera que la probabilidad de perder un paquete es independiente del estado de recepción 

(recibido/perdido) del paquete anterior. 

Cuadro 1/G.107 − Ejemplos provisionales del factor de mejora A 

Ejemplo de sistema de comunicación Valor máximo de A 

Convencional (alámbrico) 0 
Movilidad mediante redes celulares en un edificio 5 
Movilidad en una zona geográfica o en un vehículo en movimiento 10 
Conexión con lugares de difícil acceso, por ejemplo, mediante 
conexiones de múltiples saltos por satélite 

20 

Cabe destacar que los valores del cuadro 1, tomados de la Rec. UIT-T G.113 [4], son sólo 
provisionales. La utilización del factor A y la selección de su valor para una aplicación específica es 
una decisión del planificador. No obstante, los valores del cuadro 1 se han de considerar como 
límites superiores absolutos para A. 

3.7 Valores por defecto 
El cuadro 2 enumera los valores por defecto para todos los parámetros de entrada utilizados en el 
algoritmo del modelo E. Se recomienda encarecidamente la utilización de estos valores por defecto 
para todos los parámetros que no son modificados durante el cálculo para la planificación. Si se 
fijan todos los parámetros a los valores por defecto, el resultado del cálculo es una calidad muy alta 
(factor de evaluación R = 93,2). 

Cuadro 2/G.107 − Valores por defecto y gamas permitidas para los parámetros 

Parámetro Abre-
viatura Unidad Valor por 

defecto 
Gama 

permitida Comentarios 

Índice de sonoridad en emisión SLR dB  +8 0 ... +18 (Nota 1) 
Índice de sonoridad en recepción RLR dB  +2 −5 ... +14 (Nota 1) 
Índice de enmascaramiento para el efecto 
local 

STMR dB  15 10 ... 20 (Nota 2) 

Índice de efecto local para el oyente LSTR dB  18 13 ... 23 (Nota 2) 
Valor D del teléfono, lado emisor Ds –  3 –3 ... +3 (Nota 2) 
Valor D del teléfono, lado recepción Dr –  3 −3 ... +3 (Nota 2) 
Índice de sonoridad del eco para el 
hablante 

TELR dB  65 5 ... 65  

Pérdida de trayecto de eco ponderado WEPL dB  110 5 ... 110  
Retardo medio en un sentido del trayecto 
de eco 

T ms  0 0 ... 500  

Retardo de ida y vuelta en un bucle a 
4 hilos 

Tr ms  0 0 ... 1000  

Retardo absoluto en conexiones sin eco  Ta ms  0 0 ... 500  
Número de unidades de distorsión de 
cuantificación 

qdu –  1 1 ... 14  

Factor de degradación de equipo  Ie –  0 0 ... 40  

Factor de robustez contra pérdida de 
paquetes 

Bpl –  1 1 ... 40 (Nota 3) 

Probabilidad de pérdida de paquetes 
aleatoria 

Ppl %  0 0 ... 20 (Nota 3) 

Relación de ráfaga BurstR –  1 1 … 2 (Nota 3) 

Cuadro 2/G.107 − Valores por defecto y gamas permitidas para los parámetros 

Parámetro Abre-
viatura Unidad Valor por 

defecto 
Gama 

permitida Comentarios 

Ruido de circuito referido al punto de 
0 dBr  

Nc dBm0p  −70 −80 ... −40  

Nivel de ruido en el lado recepción  Nfor dBmp  −64 – (Nota 3) 
Ruido ambiente en el lado emisor  Ps dB(A)  35 35 ... 85  
Ruido ambiente en el lado recepción  Pr dB(A)  35 35 ... 85  
Factor de mejora  A –  0 0 ... 20  
NOTA 1 – Valores totales entre el micrófono o el receptor y el punto de 0 dBr. 
NOTA 2 – Relación fija: LSTR = STMR + D. 
NOTA 3 – Actualmente en estudio. 

La revisión de esta Recomendación efectuada en 2000 proporcionó una versión mejorada del 
algoritmo del modelo E (véase el anexo A). 

Debido a esta revisión del año 2000, el índice R resultante, con todos los valores por defecto de los 
parámetros, ha variado ligeramente (de R = 94,2 a R = 93,2). No obstante, a los efectos prácticos de 
la planificación, esta ligera desviación debe considerarse insignificante. 

 

Anexo A 
 

Condiciones de uso del modelo E 
 
NOTA – La evaluación y mejora del algoritmo del modelo E está en estudio. Los resultados se incluirán tan 
pronto estén disponibles. 

A.1 Ejemplos de condiciones que requieren precaución cuando se utiliza el modelo E 
– Nivel global de los factores de degradación del equipo 
 Algunas investigaciones experimentales sugieren que la tendencia general de los factores de 

degradación del equipo es demasiado pesimista, por lo que se puede incorporar un margen 
de seguridad oculto. 

– Propiedad de aditividad global del modelo 
 El modelo E presupone que diferentes tipos de degradaciones son aditivos en la escala del 

factor de determinación de índices de transmisión R. Esta característica no se ha 
comprobado de forma satisfactoria. En particular, se dispone de muy pocas investigaciones 
respecto a la interacción de códecs de baja velocidad binaria con otros tipos de 
degradaciones, por ejemplo, con ruido ambiente. Además, permanecen inciertos los efectos 
de orden de la puesta en cascada de varios códecs de baja velocidad binaria. 

– Cobertura del efecto local del hablante 
 Algunos experimentos muestran que el modelo E descarta algunos efectos de 

enmascaramiento que se producen en el efecto local del hablante, a saber, junto con el ruido 
de circuito, el ruido ambiente en el lado recepción y el eco para el hablante de retardo bajo 
(< 10 ms). 

– Factor de mejora A 
 Hasta la fecha no se ha aclarado en qué condiciones deben aplicarse los valores dados para 

el factor de mejora. Se supone que estos valores pueden depender, por ejemplo, del grupo 
de usuarios, y que los valores absolutos cambiarán a largo plazo. 

– Metodología para obtener los  factores de degradación de equipos nuevos 
 La Rec. UIT-T P.833 [6] contiene la nueva metodología adoptada para obtener los factores 

de degradación de los equipos a partir de pruebas subjetivas de calidad de escucha y la 
Rec. UIT-T P.834 [7] contiene la nueva metodología adoptada para obtener los factores de 
degradación de los equipos a partir de modelos instrumentales, tales como los descritos en 
la Rec. UIT-T P.862 [8]. 

– Predicciones para diferentes tipos de ruido ambiente y diferentes conformaciones de 
frecuencia en el canal de comunicación, en el trayecto de efecto local y en el trayecto 
de eco 

 El modelo E considera el efecto del ruido ambiente solo mediante un nivel ponderado A. La 
opinión real sobre la calidad de la comunicación vocal puede depender incluso del tipo y de 
la perturbación del ruido ambiente. Las características de frecuencia del canal de 
comunicación, del trayecto de efecto local y del eco no se consideran explícitamente en el 
modelo E, sino sólo de forma implícita mediante los índices de sonoridad. Sin embargo, 
pueden afectar a la calidad de transmisión percibida. 

A.2 Condiciones para las que se ha mejorado el modelo E actualizando la versión anterior 
– Efecto del ruido ambiente en el lado emisor 
 Con el actual algoritmo del modelo E mejorado (revisión del año 2000), ahora se tiene en 

cuenta el efecto Lombard (es decir, el hecho de que el hablante adapta su pronunciación y 
nivel de voz al entorno de ruido), que se pasó por alto en la versión de 1998 y que condujo 
a predicciones del modelo E demasiado pesimistas para altos niveles de ruido ambiente, Pr. 

– Predicciones para la distorsión de cuantificación 
 En el caso de la versión de 1998 del modelo E, los resultados de pruebas subjetivas para las 

condiciones de la unidad de referencia de ruido modulado (MNRU) muy a menudo eran 
más pesimistas que las predicciones del modelo E. Los gráficos de la figura A.1 se han 
obtenido a partir de la versión de 1998 y de la revisión del año 2000 del modelo E con los 
demás parámetros en sus valores por defecto. 

G.107_FA.1

Figura A.1/G.107 − Relación entre el número de qdu y el índice R del modelo E 

 Con respecto al algoritmo ligeramente mejorado del modelo E que se da en la presente 
Recomendación, se ha cambiado la relación entre el parámetro qdu y el índice R del modelo 
E para alinear mejor el algoritmo con los resultados de pruebas subjetivas disponibles. 

− Predicciones para el funcionamiento del códec en condiciones de pérdida de paquetes 
aleatoria 

 Las degradaciones debidas a los códecs en condiciones de pérdida de paquetes fueron 
tratadas anteriormente utilizando factores de degradación del equipo tabulados 
dependientes del códec para diferentes tasas de pérdidas de paquetes (en versiones 
anteriores del apéndice I/G.113). Como la finalidad declarada es reducir la cantidad de 
datos tabulados para uso con el modelo E, se ha investigado las posibilidades de sustituir 
los factores Ies tabulados para pérdida de paquetes por las correspondiente fórmulas. El 
método elegido produce resultados muy similares a los definidos previamente como Ie para 
todos los códecs tratados en la versión de 2001 del apéndice I/G.113.  

 Predicciones para la calidad de funcionamiento del códec en condiciones de pérdida de 
paquetes dependiente 

 A la vista de esta versión del algoritmo, en el modelo E se han integrado las distribuciones 
de pérdida caracterizadas por dependencias de las pérdidas a mediano plazo (corto plazo), 
caso que es el contrario al que se caracteriza por dependencia de las pérdidas a largo plazo. 
Hasta el presente, el enfoque incluido se ha evaluado únicamente para el códec G.729(A), 
pero se supone que podría aplicarse también al códec G.723.1 y también a otros. A reserva 
de mayor verificación, el algoritmo no debe utilizarse con relaciones de pérdida superiores 
a BurstR = 2,0. Asimismo, el modelo podría aplicarse a relaciones de pérdida superiores a 
2,0, si los porcentajes de pérdida de paquetes, Ppl, son inferiores a un 2%. 

− El efecto local del hablante 
 Las estimaciones de la calidad vocal en función de STMR para valores >15 dB calculadas 

de acuerdo con la versión anterior de la Rec. UIT-T G.107 (07/2002) son demasiado 
pesimistas y no corresponden con exactitud a los resultados de las pruebas de audición. 
Esto es especialmente importante para los teléfonos de América del Norte, los cuales 
normalmente tienen especificado un valor nominal de STMR comprendido entre 16 y 
18 dB. 

 En la presente revisión del algoritmo para el modelo E, esto queda recogido en la 
modificación de la correspondiente fórmula del factor Ist como función del efecto local 
(STMR), véase la ecuación 3-11. 

 Según se dice en el cuerpo principal de esta Recomendación, el eco para el hablante es más 
perceptible a valores de STMR correspondientes al silencio. Para solucionar esto Idte se 
sustituye por Idtes, ecuación 3-24. Para mantener la coherencia, el umbral del eco para el 
hablante para STMR > 15 dB (G.107, 07/2002) se amplió a STMR > 20 dB (G.107, versión 
revisada). Las modificaciones no afectan a los valores de STMR < 15 dB. Por consiguiente, 
la predicción de la calidad para el factor de velocidad de la transmisión R en la 
configuración por defecto (STMR = 15 dB) no difiere de la correspondiente a la versión 
anterior del modelo (07/2002). El valor por defecto de R es 93,2 para las dos versiones. En 
la figura A.2 se ilustra lo anterior. 

Figura A.2/G.107 − Comparación de R en función 
de STMR entre la versión actual y la anterior 

del algoritmo del modelo E 

 

Anexo B 
 

Medidas de calidad obtenidas a partir del factor 
de determinación de índices de transmisión R 

El factor de determinación de índices de transmisión R puede variar entre 0 y 100, donde R = 0 
representa una calidad extremadamente mala y R = 100 representa una calidad muy alta. El 
modelo E proporciona una estimación estadística de las medidas de calidad. Los porcentajes para 
una estimación buena o mejor (GoB, good or better) o mediocre o peor (PoW, poor or worse) se 
obtienen del factor R mediante la función de error Gaussiana: 

  ∫
∞π

=
x t

dtxE
–

2
–

2

e
2
1)(  (B-1) 

Las ecuaciones son: 

Se puede obtener una estimación de las notas medias de opinión (MOSCQE) en la escala de 1 a 5 a 
partir del factor R utilizando las fórmulas: 

Para R < 0: 1MOSCQE =  

Para 0 < R < 100: 6
CQE 107)100)(60(035,01MOS −⋅−−++= RRRR  (B-4) 

Para R > 100: 5,4MOSCQE =  

Esta fórmula puede ser invertida en la gama 6,5 ≤ R ≤ 100 para calcular R a partir de MOSCQE, 
véase el apéndice I. Las figuras B.1 y B.2 describen respectivamente GoB, PoW y MOSCQE en 
función de R. 

Figura B.1/G.107 − GoB (buena o mejor) y PoW (mediocre o peor) 
en función del factor de determinación de índices R 

Figura B.2/G.107 − MOSCQE en función del factor 
de determinación de índices R 

Es probable que, en algunos casos, los planificadores de sistemas de transmisión no estén 
familiarizados con la utilización de medidas de calidad tales como el factor de determinación de 
índices R obtenido a partir de cálculos de planificación y, por este motivo, en el cuadro B.13 figura 
una orientación provisional para interpretar los factores R calculados a efectos de planificación. Este 
cuadro contiene también valores de R equivalentes transformados en valores MOSCQE, GoB y PoW 
de conversación. 

Cuadro B.1/G.107 − Orientación provisional para la relación 
entre el valor R y la satisfacción del usuario 

Valor R  
(límite inferior) 

MOSCQE 
(límite inferior) 

GoB(%) 
(límite inferior)

PoW(%) 
(límite superior) Satisfacción del usuario 

90 4,34 97  ∼0 Muy satisfecho 

80 4,03 89  ∼0 Satisfecho 

70 3,60 73  6 Algunos usuarios insatisfechos
60 3,10 50  17 Muchos usuarios insatisfechos 
50 2,58 27  38 Casi todos los usuarios 

insatisfechos 
 

Anexo C 
 

Código fuente para G.107_5 en BASIC 
 

1  CLS 
2  PRINT "PROGRAM g107_4" 
3  REM THIS VERSION IS CONFORM WITH THE ALGORITHM 
4  REM DESCRIBED IN REC. G.107 
5  REM PROGRAM WRITTEN BY N.O. JOHANNESSON 
6  REM MODIFIED BY S. MOELLER, 1999; A. RAAKE, 2003, 2005 
7  PRINT 
8  PRINT "E-model, algorithm according to ITU-T Rec. G.107 (2003) Annex C," 
9  PRINT "for voice communication between side (S) and (R)." 
10 PRINT 
11 PRINT "The E-model gives an estimation of voice transmission quality" 
12 PRINT "mouth-to-ear as perceived at Side (R). The model output is a" 
13 PRINT "transmission rating factor R which can be transformed to give" 
14 PRINT "estimates of customer opinion. Such estimates are only made for" 
15 PRINT "transmission planning purposes and not for actual customer" 
16 PRINT "opinion prediction. See ITU-T Rec. G.107 (2003) for more details." 
17 PRINT 
18 PRINT "Note: To continue program:" 
19 PRINT "a) INPUT appropriate value as indicated on screen," 
20 PRINT "or" 
21 PRINT "b) if no value is asked for, PRESS THE RETURN KEY." 
22 PRINT 
23 DIM X(100), Y(5, 100), Col(5), Pxo(5) 
24 REM Set parameters at default values 
25 GOSUB 29 
26 REM Goto main menu 
27 GOTO 199 
 

____________________ 
3  La fuente del cuadro B.1 es el cuadro 1/G.109 [3]. 

29 REM Default Parameter Values, according to Tab.3/G.107 (lines 29-60) 
30 SLR = 8 
31 RLR = 2 
32 STMRs = 15 
33 Ds = 3 
34 STMR = 15 
35 Dr = 3 
36 LSTRr = 18 
37 TELR = 65 
38 T = 0 
39 WEPL = 110 
40 Tr = 0 
41 Ta = 0 
42 Ie = 0 
43 Bpl = 1 
44 Ppl = 0 
45 BurstR = 1 
46 A = 0 
47 Nc = -70 
48 Ps = 35 
49 Pr = 35 
50 qdu = 1 
55 Nfor = -64 
60 RETURN 
 
 
198 REM Main menu (lines 198-370) 
199 PRINT 
200 PRINT "Print current parameter values ", "=1" 
210 PRINT "Input new parameter values     ", "=2" 
220 PRINT "Compute table, one parameter   ", "=3" 
230 PRINT "Set parameter at default values", "=4" 
240 PRINT "Exit program                   ", "=5" 
250 PRINT 
260 INPUT Y1 
270 CLS 
280 IF Y1 = 1 THEN GOSUB 500 
290 IF Y1 = 2 THEN GOSUB 1000 
300 IF Y1 = 3 THEN GOSUB 2000 
310 IF Y1 = 4 THEN GOSUB 30 
320 IF Y1 = 5 THEN GOTO 9999 
330 CLS 
340 IF Y1 = 4 THEN PRINT , "Parameters set at default values !" 
350 GOTO 199 
 
500 REM SUB Print current parameter values (lines 500-700) 
510 PRINT , "SLR="; SLR, "RLR="; RLR, "OLR= SLR + RLR="; SLR + RLR 
520 PRINT , "Side (S): Ds="; Ds 
530 PRINT , "Side (R): STMR="; STMR, "Dr="; Dr, "LSTR="; STMR + Dr 
540 PRINT 
550 PRINT , "TELR="; TELR, "Mean One-way Delay T ms="; T 
560 PRINT , "WEPL="; WEPL, "Round-trip Delay Tr ms="; Tr 
570 PRINT , "One-way Absolute Delay Ta ms="; Ta 
580 PRINT 
590 PRINT , "Noise Floor at Side (R) Nfor dBmp="; Nfor 
600 PRINT , "Circuit Noise Nc dBm0p="; Nc 
610 PRINT , "Room Noise, Side (S), Ps dB(A)="; Ps 
620 PRINT , "Room Noise, Side (R), Pr dB(A)="; Pr 
630 PRINT 
640 PRINT , "qdu="; qdu 
650 PRINT 
660 PRINT , "Equipment Impairment Factor Ie="; Ie 
661 PRINT 

662 PRINT , "Packet-loss Robustness Factor Bpl="; Bpl 
663 PRINT 
664 PRINT , "Packet-loss Rate Ppl % ="; Ppl 
665 PRINT 
666 PRINT , "Burst Ratio ="; BurstR 
667 PRINT 
670 PRINT , "Advantage Factor A="; A 
680 PRINT 
690 INPUT C$ 
700 RETURN 
 
1000 REM SUB Input Parameters (lines 1000-1270) 
1020 CLS 
1030 PRINT "Type designation of parameter for which the value is to be changed 
!" 
1031 PRINT 
1032 PRINT "Note 1. New value of OLR is obtained indirectly, i.e. by new" 
1033 PRINT "value of SLR or RLR. (OLR=SLR+RLR.)" 
1034 PRINT 
1035 PRINT "Note 2. New value of LSTR is obtained indirectly, i.e. by new" 
1036 PRINT "value of STMR or Dr. (LSTR=STMR+Dr.)" 
1037 PRINT 
1040 INPUT "Parameter:"; A$ 
1050 INPUT "New Value="; Px 
1060 PRINT A$; "="; Px 
1070 IF ((A$ = "SLR") OR (A$ = "slr") OR (A$ = "Slr")) THEN SLR = Px 
1080 IF ((A$ = "RLR") OR (A$ = "rlr") OR (A$ = "Rlr")) THEN RLR = Px 
1090 IF ((A$ = "STMR") OR (A$ = "stmr") OR (A$ = "Stmr")) THEN STMR = Px 
1100 IF ((A$ = "Dr") OR (A$ = "DR") OR (A$ = "dr")) THEN Dr = Px 
1110 IF ((A$ = "Ds") OR (A$ = "DS") OR (A$ = "ds")) THEN Ds = Px 
1120 IF ((A$ = "TELR") OR (A$ = "telr") OR (A$ = "Telr")) THEN TELR = Px 
1130 IF ((A$ = "T") OR (A$ = "t")) THEN T = Px 
1140 IF ((A$ = "WEPL") OR (A$ = "wepl") OR (A$ = "Wepl")) THEN WEPL = Px 
1150 IF ((A$ = "Tr") OR (A$ = "TR") OR (A$ = "tr")) THEN Tr = Px 
1160 IF ((A$ = "Ta") OR (A$ = "TA") OR (A$ = "ta")) THEN Ta = Px 
1170 IF ((A$ = "Ie") OR (A$ = "IE") OR (A$ = "ie")) THEN Ie = Px 
1171 IF ((A$ = "Bpl") OR (A$ = "BPL") OR (A$ = "bpl")) THEN Bpl = Px 
1172 IF ((A$ = "Ppl") OR (A$ = "PPL") OR (A$ = "ppl")) THEN Ppl = Px 
1173 IF ((A$ = "BurstR") OR (A$ = "BURSTR") OR (A$ = "burstr")) THEN BurstR = Px 
1180 IF ((A$ = "A") OR (A$ = "a")) THEN A = Px 
1190 IF ((A$ = "Nc") OR (A$ = "NC") OR (A$ = "nc")) THEN Nc = Px 
1200 IF ((A$ = "Ps") OR (A$ = "PS") OR (A$ = "ps")) THEN Ps = Px 
1210 IF ((A$ = "Pr") OR (A$ = "PR") OR (A$ = "pr")) THEN Pr = Px 
1220 IF ((A$ = "qdu") OR (A$ = "QDU") OR (A$ = "Qdu")) THEN qdu = Px 
1230 IF ((A$ = "Nfor") OR (A$ = "NFOR") OR (A$ = "nfor")) THEN Nfor = Px 
1240 PRINT 
1250 IF Y1 = 2 THEN INPUT "More parameters changed, Yes(1) or No(0)"; Ypar 
1260 IF Ypar = 1 THEN GOTO 1020 
1270 RETURN 
 
2000 REM SUB Tabulate (lines 2000-3000) 
2020 INPUT "Variable Parameter:"; A$ 
2030 PRINT "(To exit tabulation, put parameter value = 1000 !)" 
2040 PRINT TAB(8); A$; TAB(18); "R"; TAB(28); "GOB %"; TAB(38); "POW %"; 
TAB(48); "MOS" 
2050 INPUT Px 
2060 IF Px = 1000 THEN GOTO 3000 
2070 IF ((A$ = "SLR") OR (A$ = "slr") OR (A$ = "Slr")) THEN SLR = Px 
2080 IF ((A$ = "RLR") OR (A$ = "rlr") OR (A$ = "Rlr")) THEN RLR = Px 
2090 IF ((A$ = "STMR") OR (A$ = "stmr") OR (A$ = "Stmr")) THEN 
2100   STMR = Px 
2110   LSTR = STMR + Dr 
2120 END IF 
2130 IF ((A$ = "Dr") OR (A$ = "DR") OR (A$ = "dr")) THEN 

2140   Dr = Px 
2150   LSTR = STMR + Dr 
2160 END IF 
2170 IF ((A$ = "TELR") OR (A$ = "telr") OR (A$ = "Telr")) THEN TELR = Px 
2180 IF ((A$ = "T") OR (A$ = "t")) THEN T = Px 
2190 IF ((A$ = "WEPL") OR (A$ = "wepl") OR (A$ = "Wepl")) THEN WEPL = Px 
2200 IF ((A$ = "Tr") OR (A$ = "TR") OR (A$ = "tr")) THEN Tr = Px 
2210 IF ((A$ = "Ta") OR (A$ = "TA") OR (A$ = "ta")) THEN Ta = Px 
2220 IF ((A$ = "Ie") OR (A$ = "IE") OR (A$ = "ie")) THEN Ie = Px 
2221 IF ((A$ = "Bpl") OR (A$ = "BPL") OR (A$ = "bpl")) THEN Bpl = Px 
2222 IF ((A$ = "Ppl") OR (A$ = "PPL") OR (A$ = "ppl")) THEN Ppl = Px 
2223 IF ((A$ = "BurstR") OR (A$ = "BURSTR") OR (A$ = "burstr")) THEN BurstR = Px 
2230 IF ((A$ = "A") OR (A$ = "a")) THEN A = Px 
2240 IF ((A$ = "Nc") OR (A$ = "NC") OR (A$ = "nc")) THEN Nc = Px 
2245 IF ((A$ = "Nfor") OR (A$ = "NFOR") OR (A$ = "nfor")) THEN Nfor = Px 
2250 IF ((A$ = "Ps") OR (A$ = "PS") OR (A$ = "ps")) THEN Ps = Px 
2260 IF ((A$ = "Pr") OR (A$ = "PR") OR (A$ = "pr")) THEN Pr = Px 
2270 IF ((A$ = "qdu") OR (A$ = "QDU") OR (A$ = "Qdu")) THEN qdu = Px 
2280 IF ((A$ = "Ie") OR (A$ = "IE") OR (A$ = "ie")) THEN Ie = Px 
2290 IF ((A$ = "Ds") OR (A$ = "DS") OR (A$ = "ds")) THEN Ds = Px 
2300 GOSUB 3500 
2400 GOSUB 4000 
2500 GOSUB 4100 
2600 GOSUB 4200 
2700 R = INT(R * 10 + .5) / 10 
2800 PRINT TAB(8); Px; TAB(18); R; TAB(28); GOB; TAB(38); POW; TAB(48); MOS 
2900 GOTO 2050 
3000 RETURN 
 
3500 REM Compute R (lines 3500-3880) 
 
3509 REM Noise Summation, formulas (3) to (7) 
3510 Nr1 = Ps – SLR – Ds – 100 
3520 Nr1 = Nr1 + .004 * (Ps – SLR – RLR – Ds – 14) ^ 2 
3530 LSTR = STMR + Dr 
3540 Pro = Pr + 10 * LOG(1 + 10 ^ ((10 – LSTR) / 10)) / LOG(10) 
3550 Pr1 = Pro + .008 * (Pro – 35) ^ 2 
3560 Nr2 = Pr1 – 121 + RLR 
3570 Nfo = Nfor + RLR 
3580 No = 10 * LOG(10 ^ (Nr1 / 10) + 10 ^ (Nr2 / 10) + 10 ^ (Nc / 10) + 10 ^ 
(Nfo / 10)) / LOG(10) 
3590 Nt = No – RLR 
 
3599 REM Ro, formula (2) 
3600 Ro = 15 – 1.5 * (SLR + No) 
 
3609 REM Iolr, formulas (9) and (10) 
3610 Xolr = SLR + RLR + .2 * (64 + Nt) 
3620 Iolr = 20 * ((1 + (Xolr / 8) ^ 8) ^ (1 / 8) – Xolr / 8) 
 
3629 REM Ist, formulas (11) and (12) 
3630 STMRo = -10 * LOG(10 ^ (-STMR / 10) + 10 ^ (-TELR / 10) * EXP(-T / 4)) / 
LOG(10) 
3640 Ist = 12 * (1 + ((STMRo – 13) / 6) ^ 8) ^ (1 / 8) 
3645 Ist = Ist – 28 * (1 + ((STMRo + 1) / 19.4) ^ 35) ^ (1 / 35) 
3650 Ist = Ist – 13 * (1 + ((STMRo – 3) / 33) ^ 13) ^ (1 / 13) + 29 
 
3659 REM Iq, formulas (13) to (17) 
3660 IF qdu < 1 THEN qdu = 1 
3670 Q = 37 – 15 * LOG(qdu) / LOG(10) 
3680 G = 1.07 + .258 * Q + .0602 * Q ^ 2 
3690 Iq = 15 * LOG(1 + 10 ^ ((Ro – 100) / 15) * 10 ^ (46 / 8.4 – G / 9) + 10 ^ 
(46 / 30 – G / 40)) / LOG(10) 

3699 REM Is, formula (8) 
3700 Isyn = Iolr + Ist + Iq 
 
3709 REM TERV, formula (22) 
3710 TERV = TELR + 6 * EXP(-.3 * T ^ 2) – 40 * LOG((1 + T / 10) / (1 + T / 150)) 
/ LOG(10) 
3719 REM Modifications to satisfy formula (23) 
3720 IF STMR < 9 THEN TERV = TERV + .5 * Ist 
 
3729 REM Idte, formulas (19) to (21) 
3730 Re = 80 + 2.5 * (TERV – 14) 
3740 Roe = -1.5 * (No – RLR) 
3750 Xdt = (Roe – Re) / 2 
3760 Idte = Xdt + SQR(Xdt ^ 2 + 100) 
3770 Idte = (Idte – 1) * (1 – EXP(-T)) 
 
3779 REM Modifications to satisfy formula (24) 
3780 IF STMR > 20 THEN Idte = SQR(Idte ^ 2 + Ist ^ 2) 
 
3789 REM Idle, formulas (25) and (26)        
3790 Rle = 10.5 * (WEPL + 7) * (Tr + 1) ^ (-1 / 4) 
3800 Xdl = (Ro – Rle) / 2 
3810 Idle = Xdl + SQR(Xdl ^ 2 + 169) 
 
3819 REM Idd, formulas (27) and (28) 
3820 IF Ta < 100 THEN Idd = 0 
3830 IF Ta = 100 THEN Idd = 0 
3840 IF Ta > 100 THEN 
        X = (LOG(Ta / 100)) / LOG(2) 
        Idd = 25 * ((1 + X ^ 6) ^ (1 / 6) – 3 * (1 + (X / 3) ^ 6) ^ (1 / 6) + 2) 
3850 END IF 
 
3859 REM Id 
3860 Id = Idte + Idle + Idd 
 
3864 REM Inclusion of packet-loss: Ieef, formula (29) 
3865 Ieef = Ie + (95 - Ie) * (Ppl / ((Ppl / BurstR) + Bpl)) 
 
3869 REM R, formula (1) 
3870 R = Ro – Isyn – Id – Ieef + A 
3880 RETURN 
 
4000 REM Compute GOB, formula (B.2) (lines 4000-4050) 
4010 Z# = (R – 60) / 16 
4020 GOSUB 5000 
4030 GOB = 100 * F# 
4040 GOB = INT(GOB * 10 + .5) / 10 
4050 RETURN 
 
4100 REM Compute POW, formula (B.3) (lines 4100-4150) 
4110 Z# = (R – 45) / 16 
4120 GOSUB 5000 
4130 POW = 100 * (1 – F#) 
4140 POW = INT(POW * 10 + .5) / 10 
4150 RETURN 
 
4200 REM Compute MOS, formula (B.4) (lines 4200-4260) 
4210 MOS = 1 + R * .035 + R * (R – 60) * (100 – R) * 7 * 10 ^ (-6) 
4220 MOS = INT(MOS * 100 + .5) / 100 
4230 IF R < 0 THEN MOS = 1 
4240 IF MOS < 1 THEN MOS = 1 
4250 IF R > 100 THEN MOS = 4.5 
4260 RETURN 

5000 REM Norm Distr F(Z), formula (B.1) (lines 5000-5130) 
5010 S# = 0 
5020 N% = 0 
5030 H# = Z# 
5040 S# = S# + H# 
5050 H# = H# * (-1) * (Z#) ^ 2 * (2 * N% + 1) / ((N% + 1) * 2 * (2 * N% + 3)) 
5060 N% = N% + 1 
5070 IF ABS(H#) < 10 ^ (-6) THEN GOTO 5090 
5080 GOTO 5040 
5090 S# = S# / (SQR(2 * 3.14159265#)) 
5100 F# = .5 + S# 
5110 F# = INT(F# * 10 ^ 5 + .5) / 10 ^ 5 
5120 REM PRINT "Z="; Z#, "F(Z)="; F#, "N="; N% 
5130 RETURN 
9999 END 

 

Apéndice I 
 

Cálculo de R a partir de los valores de MOSCQE 

En la gama 6,5 ≤ R ≤ 100, R puede ser calculado a partir de MOSCQE usando la fórmula: 

La función arctan2(x, y) se implementa en ANSI C como la función atan2(y, x). Los usuarios deben 
observar que el orden de los dos parámetros difiere en este caso. 

